Fourier series examples

Paul

February 26, 2019

1 Summary of Fourier series stuff

Periodic Fourier series takes a function u and returns a sequence $\alpha = f(u)$ defined by

$$\alpha_k = f(u)_k = \frac{1}{2L}\int_{-L}^{L} e^{-i\frac{k\pi}{L}x}u(x)dx$$

The series is then defined by

$$\sum_{k=-\infty}^{\infty} \alpha_k e^{i\frac{k\pi}{L}x}$$

The partial sums we denote by

$$u_n(x) = \sum_{k=-n}^{n} \alpha_k e^{i\frac{k\pi}{L}x}$$

For $k \geq 1$ we write $\alpha_k = \frac{1}{2}a_k - i\frac{1}{2}b_k$, where

$$a_k = \frac{1}{L}\int_{-L}^{L} \cos\left(\frac{k\pi}{L}x\right) u(x)dx,$$
$$b_k = \frac{1}{L}\int_{-L}^{L} \sin\left(\frac{k\pi}{L}x\right) u(x)dx.$$

Note that

$$\alpha_{-k} = \frac{1}{2}a_k + i\frac{1}{2}b_k.$$

Furthermore, in the $k = 0$ case we write

$$\alpha_0 = a_0 = \frac{1}{2L}\int_{-L}^{L} u(x)\,dx.$$

Assembling these pieces together we find that

$$\begin{aligned}
u_n(x) &= a_0 + \sum_{k=1}^{n} \frac{1}{2}(a_k - ib_k)e^{i\frac{k\pi}{L}x} + \sum_{k=1}^{n} \frac{1}{2}(a_k + ib_k)e^{-i\frac{k\pi}{L}x} \\
&= a_0 + \sum_{k=1}^{n} a_k \frac{e^{i\frac{k\pi}{L}x} + e^{-i\frac{k\pi}{L}x}}{2} + \sum_{k=1}^{n} b_k \frac{e^{i\frac{k\pi}{L}x} - e^{-i\frac{k\pi}{L}x}}{2i} \\
&= a_0 + \sum_{k=1}^{n} a_k \cos\left(\frac{k\pi}{L}x\right) + \sum_{k=1}^{n} b_k \sin\left(\frac{k\pi}{L}x\right)
\end{aligned}$$

Of course we can take $n \to \infty$ to obtain the full Fourier series.

In the following examples we compute the coefficients α_k in general. Then we compute the coefficients a_k and b_k in the case that $L = \pi$. We use the cosine/sine form of the sum for plotting.

In [1]:
```
# setup for our variables
var('x','y','k','x','L','a')
assume(k, 'integer')
assume(L, 'real')
assume(L>0)
```

1.1 Example $u(x) = x$

Compute Fourier coefficients

In [2]:
```
u(x) = x

alpha(k,L) = (1/(2*L))*integrate( exp(-I*k*pi*y/L)*u(y), (y,-L,L))

show(alpha(k))
```

```
1/2*((I*pi*L^2*k - L^2)*e^(I*pi*k)/(pi^2*k^2) + (I*pi*L^2*k + L^2)*e^(-I*pi*k)/(pi^2*k^2))/L
```

With a little algebra, we are able to simplify this to

$$\alpha_k = (-1)^k \frac{iL}{k\pi} \qquad k \neq 0.$$

For α_0 we compute

In [3]:
```
alpha0 = (1/(2*L))*integrate(u(y), (y,-L,L))
alpha0
```

Out[3]: 0

Now we plot the partial sum f_n of the Fourier approximation. We do this using the cosine/sine version.

In [4]:
```
n=10 # fix number of terms in sum

# compute coefficients
a(k) = (1/pi)*integrate(cos(k*y)*u(y), (y,-pi, pi))
b(k) = (1/pi)*integrate(sin(k*y)*u(y), (y,-pi, pi))
a0 = integrate(u(y), (y,-pi,pi))/(2*pi)

# construct approximation
fn(x) = a0 + sum(a(k)*cos(k*x), k, 1, n) + sum(b(k)*sin(k*x), k, 1, n)

plot(fn(x),(x,-pi, pi), figsize=[4,3])
```

Out[4]:



1.2 Square wave

Here we deal with the square wave

$$u(x) = \begin{cases} -1 & \text{if } -L < x < 0 \\ 1 & \text{if } 0 < x < L. \end{cases}$$

Because Sage does not handle piecewise functions very well, we split the integrals by hand.

```
In [5]: uL(x)=-1 # left part of u
        uR(x) = 1 # right part of u

        alpha(k,L) = (1/(2*L))*(integrate( exp(-I*k*pi*y/L)*uL(y), (y,-L,0))\
            +integrate( exp(-I*k*pi*y/L)*uR(y), (y,0,L)) )

        show(alpha(k,L))
```

```
1/2*(I*L*e^(I*pi*k)/(pi*k) + I*L*e^(-I*pi*k)/(pi*k) - 2*I*L/(pi*k))/L
```

With some algebra, we simplify this to:

$$\alpha_k = i\frac{(-1)^k - 1}{k\pi} \qquad k \neq 0.$$

For $k = 0$ we compute:

```
In [6]: alpha0 = (1/(2*L))*integrate(u(y), (y,-L,L))
        alpha0
```

```
Out[6]: 0
```

```
In [7]: n=10 # fix number of terms in sum

        # compute coefficients
        a(k) = (1/pi)*(integrate(cos(k*y)*uL(y), (y,-pi, 0)) + integrate(cos(k*y)*uR(y), (y,0, p
        b(k) = (1/pi)*(integrate(sin(k*y)*uL(y), (y,-pi, 0)) + integrate(sin(k*y)*uR(y), (y,0, p
        a0 = (1/(2*pi))*(integrate(uL(y), (y,-pi, 0)) + integrate(uR(y), (y,0, pi)))

        # construct approximation
        fn(x) = a0 + sum(a(k)*cos(k*x), k, 1, n) + sum(b(k)*sin(k*x), k, 1, n)

        # plot
        plot(fn(x),(x,-pi, pi), figsize=[4,3])
```

Out[7]:



1.3 Triangle

The "triangle" is given by

$$u(x) = \begin{cases} L + x & \text{if } -L < x < 0 \\ L - x & \text{if } 0 < x < L. \end{cases}$$

Here we deal with the complex part separately from the plotting part, as L figures in the formula for u.

```
In [8]: uL(x)=L+x # left part of u
        uR(x)=L-x # right part of u

        alpha(k,L) = (1/(2*L))*( integrate( exp(-I*k*pi*y/L)*uL(y,L), (y,-L,0)) \
                               + integrate( exp(-I*k*pi*y/L)*uR(y,L), (y,0,L)))
```

```
show(alpha(k,L))
```

```
-1/2*(L^2*e^(I*pi*k)/(pi^2*k^2) + L^2*e^(-I*pi*k)/(pi^2*k^2) - (I*pi*L^2*k + L^2)/(pi^2*k^2) + (
```

With some algebra, we simplify this to:

$$\alpha_k = \frac{L}{\pi^2 k^2}(1 - (-1)^k) \qquad k \neq 0.$$

For $k = 0$ we compute:

```
In [9]: alpha0 = (1/(2*L))*( integrate(uL(y,L), (y,-L,0)) \
                              + integrate(uR(y,L), (y,0,L)))
        alpha0
```

```
Out[9]: 1/2*L
```

```
In [10]: n=10 # fix number of terms in sum

         uL(x)=pi+x # left part of u
         uR(x)=pi-x # right part of u

         # coefficients
         a(k) = (1/pi)*(integrate(cos(k*y)*uL(y), (y,-pi, 0)) + integrate(cos(k*y)*uR(y), (y,0,
         b(k) = (1/pi)*(integrate(sin(k*y)*uL(y), (y,-pi, 0)) + integrate(sin(k*y)*uR(y), (y,0,
         a0 = (1/(2*pi))*(integrate(uL(y), (y,-pi, 0)) + integrate(uR(y), (y,0, pi)))

         # construct approximation
         fn(x) = a0 + sum(a(k)*cos(k*x), k, 1, n) + sum(b(k)*sin(k*x), k, 1, n)

         # plot
         plot(fn(x),(x,-pi, pi), figsize=[4,3])
```

```
Out[10]:
```



1.4 The "sawtooth"

The Sawtooth is given by

$$u(x) = \begin{cases} x + L & \text{if } -L < x < 0 \\ x & \text{if } 0 < x < L. \end{cases}$$

```
In [11]: uL(x)=x+L # left part of u
         uR(x)=x # right part of u

         alpha(k,L) = (1/(2*L))*( integrate( exp(-I*k*pi*y/L)*uL(y,L), (y,-L,0)) \
                                 + integrate( exp(-I*k*pi*y/L)*uR(y,L), (y,0,L)))

         show(alpha(k,L))
```

```
-1/2*(L^2*e^(I*pi*k)/(pi^2*k^2) + L^2/(pi^2*k^2) - (I*pi*L^2*k + L^2)*e^(-I*pi*k)/(pi^2*k^2) - (
```

With some algebra, we simplify this to:

$$\alpha_k = -\frac{iL(1 + (-1)^k)}{2\pi k} \qquad k \neq 0.$$

For $k = 0$ we compute:

```
In [12]: alpha0 = (1/(2*L))*( integrate( uL(y,L), (y,-L,0)) \
                                 + integrate( uR(y,L), (y,0,L)))
         show(alpha0)
```

```
1/2*L
```

```
In [13]: n=10 # fix number of terms in sum

         uL(x)=x+pi # left part of u
         uR(x)=x # right part of u

         # coefficieints
         a(k) = (1/pi)*(integrate(cos(k*y)*uL(y), (y,-pi, 0)) + integrate(cos(k*y)*uR(y), (y,0,
         b(k) = (1/pi)*(integrate(sin(k*y)*uL(y), (y,-pi, 0)) + integrate(sin(k*y)*uR(y), (y,0,
         a0 = (1/(2*pi))*(integrate(uL(y), (y,-pi, 0)) + integrate(uR(y), (y,0, pi)))

         # construct approximation
         fn(x) = a0 + sum(a(k)*cos(k*x), k, 1, n) + sum(b(k)*sin(k*x), k, 1, n)

         # plot
         plot(fn(x),(x,-pi, pi), figsize=[4,3])
```

Out[13]:



1.5 Pulse of size a

The "pulse of size a"

$$u(x) = \begin{cases} 0 & \text{if } |x| > a \\ \frac{1}{2a} & \text{if } |x| < a, \end{cases}$$

In [14]:
```
var('a')
assume(a, 'real')
assume(a>0)
u(x) = 1/(2*a)

alpha(k,L,a) = (1/(2*L))*integrate( exp(-I*k*pi*y/L)*u(x), (y,-a,a))

show(alpha(k,L,a))
```

```
1/4*(-I*L*e^(I*pi*a*k/L)/(pi*k) + I*L*e^(-I*pi*a*k/L)/(pi*k))/(L*a)
```

With some algebra, we simplify this to:

$$\alpha_k = \frac{1}{2\pi k a} \sin\left(\frac{k\pi}{L}a\right) \qquad k \neq 0.$$

For $k = 0$ we compute:

In [15]:
```
alpha0 = (1/(2*L))*integrate(u(y), (y,-a,a))
show(alpha0)
```

1/2/L

For the purpose of plotting we take $L = \pi$ and $a = \pi/10$.

In [16]:
```
n=50 # fix number of terms in sum; due to the fine scale structure, we need more terms

u(x)=10/pi

# construct coefficients
a(k) = (1/pi)*integrate(cos(k*y)*u(y), (y,-pi/10,pi/10))
b(k) = (1/pi)*integrate(sin(k*y)*u(y), (y,-pi/10,pi/10))

a0 = (1/(2*pi))*integrate(u(y), (y,-pi/10,pi/10))

# construct approximation
fn(x) = a0 + sum(a(k)*cos(k*x), k, 1, n) + sum(b(k)*sin(k*x), k, 1, n)

# plot
plot(fn(x),(x,-pi, pi), figsize=[4,3])
```

Out[16]:



In [ ]: